```
                ;*********************************************************
                ;                                                         *
                ;       PROGRAM ID:     DOUBLE D BIOS  (DDBIOS)           *
                ;                                                         *
                ;*********************************************************
                ;                                                         *
                ;       VERSION:        CP/M 2.2 8"  RELEASE 2A           *
                ;                                                         *
                ;*********************************************************
                ;                                                         *
                ;       PRESENTED BY:   JADE COMPUTER PRODUCTS            *
                ;                       4901 W. ROSECRANS BLVD.           *
                ;                       HAWTHORNE, CALIFORNIA             *
                ;                       90250,   U.S.A.                   *
                ;                                                         *
                ;********************************************** SK ***

                ;*********************************************************
                ; DECLARE CP/M 2.2 SYSTEM SIZE                             *
                ;*********************************************************

0014 =          CPM$NK  EQU     20      ;SYSTEM SIZE K BYTES.

                ;*********************************************************
                ; DOUBLE D HARDWARE PARAMETER - SYSTEM PORT ADDRESS        *
                ;*********************************************************

0043 =          D$PORT  EQU     043H    ;DOUBLE D PORT ADDRESS.

                ;*********************************************************
                ; SELECT NUMBER OF DISK DRIVES USED                        *
                ;*********************************************************

0002 =          N$DRVS  EQU     2       ;SELECT 1 TO 4 DRIVES.

                ;*********************************************************
                ; DISK OPERATING SYSTEM ADDRESSES.                         *
                ;*********************************************************

0400 =          K$B     EQU     1024            ;1K BYTE SIZE.
5000 =          CPM$SZ  EQU     CPM$NK * K$B    ;TOP SYSTEM ADDRESS.
0000 =          CPM$BS  EQU     CPM$SZ-(20*K$B) ;CP/M BIAS VALUE.

0100 =          TPA     EQU     0100H           ;ADDRESS OF TPA.
3400 =          CCP     EQU     CPM$BS+3400H    ;ADDRESS OF CCP.
3C00 =          BDOS    EQU     CPM$BS+3C00H    ;ADDRESS OF BDOS.
4A00 =          BIOS    EQU     CPM$BS+4A00H    ;ADDRESS OF BIOS
C600 =          BIOS$R  EQU     1000H-BIOS      ;DDT OFFSET 1000H LOAD.
F000 =          BOOT    EQU     0F000H          ;BOOT PROM JUMP TABLE.
0003 =          IO$LOC  EQU     0003H           ;I/O BYTE LOCATION.
0004 =          DF$LOC  EQU     0004H           ;DRIVE ASSIGN LOCATION.

                ;*********************************************************
                ; DOUBLE D SYSTEM PARAMETERS                               *
                ;*********************************************************

0000 =          IOBYTE  EQU     00000000B       ;INITIAL IOBYTE VALUE.
0000 =          DF$DRV  EQU     0               ;INITIAL DEFAULT DRV.
```

```
0080 =          SEC$SZ  EQU     0080H           ;BYTES PER SECTOR.
0100 =          FMT$SZ  EQU     0100H           ;FORMAT BUFF SIZE.

                ;*********************************************************
                ; DOUBLE D HARDWARE COMMANDS                              *
                ;*********************************************************

0001 =          DC$SIN  EQU  00000001B  ;SWITCH DD BANK 0 INTO SYSTEM.
0001 =          DC$MB0  EQU  00000001B  ;SELECT DOUBLE D BANK 0.
0003 =          DC$MB1  EQU  00000011B  ;SELECT DOUBLE D BANK 1.
0000 =          DC$SOT  EQU  00000000B  ;SWITCH DD MEM OUT OF SYSTEM.
0002 =          DC$INT  EQU  00000010B  ;ISSUE DD Z80A INTERRUPT.

                ;*********************************************************
                ; DISK CONTROLLER MODULE LINKAGE (DCM - VER 2.2)          *
                ;*********************************************************

                ;*******( DCM ADDRESSES DEFINED )***********************

0370 =          DD$CBT  EQU     0370H   ;COMMAND BLOCK  (BANK 0).
0380 =          DD$BUF  EQU     0380H   ;SECTOR BUFFER  (BANK 0).
0300 =          DD$FBF  EQU     0300H   ;FORMAT BUFFER  (BANK 1).
03A0 =          DD$DPB  EQU     03A0H   ;ID SEC DPB     (BANK 0).
03B1 =          DD$DDF  EQU     03B1H   ;ID SEC FLAGS   (BANK 0).

                ;*******( DCM COMMANDS )******************************

0000 =          DC$LOG  EQU     000H    ;LOG ON DISKETTE.
0001 =          DC$RDS  EQU     001H    ;READ SECTOR.
0002 =          DC$WRS  EQU     002H    ;WRITE SECTOR.
0003 =          DC$FMT  EQU     003H    ;FORMAT TRACK.
0005 =          DC$LST  EQU     005H    ;LIST CHARACTER.
0006 =          DC$LCK  EQU     006H    ;LIST STATUS.

                ;*********************************************************
                ; ASSEMBLER DIRECTIVES                                    *
                ;*********************************************************

4A00                    ORG     BIOS

                ;*********************************************************
                ; BIOS JUMP VECTOR TABLE                                  *
                ;*********************************************************

4A00 C3364A             JMP     INIT            ;COLD START ENTRY
4A03 C3424A             JMP     WARM            ;RELOAD CCP/BDOS
4A06 C306F0             JMP     CNS$CK          ;GET CONSOLE STATUS
4A09 C309F0             JMP     CNS$IN          ;CONSOLE INPUT
4A0C C30CF0             JMP     CNS$OT          ;CONSOLE OUTPUT
4A0F C3B04A             JMP     LIST            ;PRINTER OUTPUT
4A12 C3AF4A             JMP     PUNCH           ;PUNCH OUTPUT
4A15 C3AC4A             JMP     READER          ;READER INPUT
4A18 C3D04A             JMP     HOME            ;HOME SELECTED DRIVE
4A1B C3D54A             JMP     SELDSK          ;SELECT DISK DRIVE
4A1E C3F44A             JMP     SETTRK          ;SET TRACK NUMBER
4A21 C3F94A             JMP     SETSEC          ;SET SECTOR NUMBER
4A24 C3FE4A             JMP     SETDMA          ;SET TRANSFER ADDRESS
4A27 C3044B             JMP     DISKRD          ;PERFORM DISK READ
4A2A C3244B             JMP     DISKWR          ;PERFORM DISK WRITE
4A2D C3C04A             JMP     LISTST          ;RETURN LIST STAT
```

```
4A30 C3514B            JMP     SECTRN          ;TRANSLATE SECTOR
4A33 C3614B            JMP     FORMAT          ;FORMAT A TRACK

                       ;*********************************************************
                       ; COLD START ENTRY - ISSUE SIGN ON MESSAGE                *
                       ;*********************************************************

4A36 318000    INIT:   LXI     SP,0080H        ;SET UP STACK AREA.
4A39 21964C            LXI     H,MSG$SO        ;SIGN ON MSG ADDR.
4A3C CD4C4C            CALL    MSG$OT          ;ISSUE MESSAGE.
4A3F C3574A            JMP     CPM$LD          ;LOAD CCP/BDOS.

                       ;*********************************************************
                       ; WARM BOOT ENTRY - LOADS CCP/BDOS - INITIALIZES          *
                       ;*********************************************************

                       ;*******( SET UP FOR CCP/BDOS LOAD )*******************

4A42 3A0300    WARM:   LDA     IO$LOC          ;GET I/O BYTE VALUE.
4A45 32914C            STA     IO$IMG          ;STORE I/O VALUE.
4A48 3A0400            LDA     DF$LOC          ;GET DEFAULT DRIVE.
4A4B FE02              CPI     N$DRVS          ;CHECK LEGAL DRIVE.
4A4D DA514A            JC      WRM$OK          ;IF LEGAL, GO OK.
4A50 AF                XRA     A               ;SET DRIVE TO A.
4A51 32924C    WRM$OK: STA     DF$IMG          ;STORE IN IMAGE.
4A54 318000            LXI     SP,0080H        ;SET UP STACK.
4A57 3E00      CPM$LD: MVI     A,DF$DRV        ;INIT DEFAULT DRIVE.
4A59 32594C            STA     BT$DRV          ;SELECT DISK.
4A5C 010034            LXI     B,CCP           ;CP/M CCP ADDRESS.
4A5F CDFE4A            CALL    SETDMA          ;SET DMA ADDR.
4A62 0E02              MVI     C,2             ;CCP 1ST SECTOR.
4A64 CDF94A            CALL    SETSEC          ;SET SECTOR NMBR.
4A67 0E01              MVI     C,1             ;CCP/BDOS TRACK.
4A69 CDF44A            CALL    SETTRK          ;SET TRACK NUMBER.

                       ;*******( LOAD CCP/BDOS )*****************************

4A6C CD044B    W$READ: CALL    DISKRD          ;READ ONE SECTOR.
4A6F A7                ANA     A               ;SET FLAGS.
4A70 C28C4A            JNZ     W$EROR          ;EXIT IF ERROR.
4A73 3A5B4C            LDA     BT$SEC          ;GET SECTOR NMBR.
4A76 FE2D              CPI     45              ;LAST SECTOR CHECK.
4A78 CA934A            JZ      W$ZRPG          ;GOTO ZERO PAGE SET.
4A7B 3C                INR     A               ;INCREMENT SECTOR.
4A7C 325B4C            STA     BT$SEC          ;STORE NEXT SECTOR.
4A7F 118000            LXI     D,SEC$SZ        ;GET SECTOR SIZE.
4A82 2A604C            LHLD    BT$DMA          ;GET TRANSFER ADDR.
4A85 19                DAD     D               ;CALCULATE NEW ADDR.
4A86 22604C            SHLD    BT$DMA          ;SET NEW ADDRESS.
4A89 C36C4A            JMP     W$READ          ;DO ANOTHER WARM READ.

                       ;*******( READ ERROR DETECTED )***********************

4A8C 21CB4C    W$EROR: LXI     H,MSG$LE        ;GET ERROR MESAAGE.
4A8F CD4C4C            CALL    MSG$OT          ;ISSUE MESSAGE.
4A92 76                HLT                     ;OR GOTO MONITOR

                       ;*******( INITIALIZE SYSTEM PARAMETERS )***************

4A93 010800    W$ZRPG: LXI     B,8             ;BASE IMAGE SIZE.
```

```
4A96 110000             LXI     D,0             ;BASE ADDRESS SET.
4A99 218E4C             LXI     H,BS$IMG        ;BASE IMAGE ADDR.
4A9C CD414C             CALL    BLOCK           ;BLOCK MOVE ROUTINE.
4A9F 218000             LXI     H,0080H         ;DEFAULT SECTOR BUFF.
4AA2 22604C             SHLD    BT$DMA          ;SET TRANSFER ADDRESS.

                ;*******( JUMP TO CCP )*********************************

4AA5 3A0400             LDA     DF$LOC          ;GET CURRENT DSK NMBR.
4AA8 4F                 MOV     C,A             ;SEND TO THE CCP.
4AA9 C30034             JMP     CCP             ;JUMP INTO CCP CP/M.

                ;************************************************************
                ; CONSOLE LINKAGE DEFINITIONS - BOOT PROM ADDRESSES          *
                ;************************************************************

F006 =          CNS$CK  EQU     BOOT+006H       ;CHECK CONSOLE INPUT.
F009 =          CNS$IN  EQU     BOOT+009H       ;READ CONSOLE INPUT.
F00C =          CNS$OT  EQU     BOOT+00CH       ;CHARACTER TO CONSOLE.

                ;************************************************************
                ; READER AND PUNCH DRIVERS - USER SHOULD DEFINED             *
                ;************************************************************

4AAC 3E1AC9     READER: MVI   A,CNTL$Z!RET      ;RETURN END OF FILE.
4AAF C9         PUNCH:  RET                     ;NOT IMPLEMENTED.

                ;************************************************************
                ; PRINTER DRIVER AREA - DCM SERIAL PORT LINKAGE              *
                ;************************************************************

4AB0 79         LIST:   MOV     A,C             ;LIST CHAR TO ACUM.
4AB1 325D4C             STA     BT$CHR          ;STORE LIST CHARACTER.
4AB4 3E01               MVI     A,DC$SIN        ;LOAD SWITCH MEM CMND.
4AB6 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4AB8 3E05               MVI     A,DC$LST        ;DCM LIST COMMAND.
4ABA CD8A4B             CALL    DSK$EX          ;CALL DISK EXECUTE.
4ABD C3444B             JMP     DSK$OK          ;RETURN TO CALLER.

4AC0 3E01       LISTST: MVI     A,DC$SIN        ;LOAD SWITCH MEM CMND.
4AC2 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4AC4 3E06               MVI     A,DC$LCK        ;DCM LIST STAT CMND.
4AC6 CD8A4B             CALL    DSK$EX          ;CALL DISK EXECUTE.
4AC9 CD444B             CALL    DSK$OK          ;SWITCH DD MEM OUT.
4ACC 3A5F4C             LDA     BT$STS          ;LOAD RETURN STATUS.
4ACF C9                 RET                     ;RETURN TO CALLER.

                ;************************************************************
                ; HOME - SET TRACK TO ZERO                                   *
                ;************************************************************

4AD0 0E00       HOME:   MVI     C,0             ;C REGISTER TO ZERO.
4AD2 C3F44A             JMP     SETTRK          ;PERFORM SET TRACK.

                ;************************************************************
                ; SELECT DISK DRIVE - CHECK FOR LOGON                        *
                ;************************************************************

4AD5 210000     SELDSK: LXI     H,0             ;ERROR RETURN CODE.
4AD8 79                 MOV     A,C             ;PUT DRIVE NMBR IN A.
```

```
4AD9 FE02              CPI     N$DRVS          ;CHECK IF LEGAL DRIVE.
4ADB D0                RNC                     ;NO CARRY IF ILLEGAL.
4ADC 32594C            STA     BT$DRV          ;STORE DRIVE NUMBER.
4ADF 7B                MOV     A,E             ;CHECK IF LOG-ON REQ.
4AE0 32644C            STA     LOG$RQ          ;STORE LOGON REGISTER.
4AE3 3A594C    RETDSK: LDA     BT$DRV          ;GET DRIVE NUMBER.
4AE6 6F                MOV     L,A             ;L SET DISK NUMBER.
4AE7 2600              MVI     H,0             ;ZERO H REGISTER.
4AE9 29                DAD     H               ;*2.
4AEA 29                DAD     H               ;*4.
4AEB 29                DAD     H               ;*8.
4AEC 29                DAD     H               ;*16 (SIZE OF HEADER).
4AED 11E94C            LXI     D,D0$DPH        ;DRIVE 0 D$P$H.
4AF0 19                DAD     D               ;HLSET DRIVE N DPH.
4AF1 C3B54B            JMP     LOG$ON          ;GO CHECK LOG-ON.

               ;*************************************************
               ; SET TRACK NUMBER                                *
               ;*************************************************

4AF4 79        SETTRK: MOV     A,C             ;MOVE TRACK NUMBER.
4AF5 325A4C            STA     BT$TRK          ;SAVE TRACK NUMBER.
4AF8 C9                RET                     ;RETURN TO CALLER.

               ;*************************************************
               ; SET SECTOR NUMBER                               *
               ;*************************************************

4AF9 79        SETSEC: MOV     A,C             ;MOVE SECTOR NUMBER.
4AFA 325B4C            STA     BT$SEC          ;SAVE SECTOR NUMBER.
4AFD C9                RET                     ;RETURN TO CALLER.

               ;*************************************************
               ; SET MEMORY ADDRESS FOR DISK TRANSFER            *
               ;*************************************************

4AFE 60        SETDMA: MOV     H,B             ;HIGH ORDER MOVE.
4AFF 69                MOV     L,C             ;LOW ORDER MOVE.
4B00 22604C            SHLD    BT$DMA          ;SAVE TRANSFER ADDRESS.
4B03 C9                RET                     ;RETURN TO CALLER.

               ;*************************************************
               ; READ A DISK SECTOR ROUTINE                      *
               ;*************************************************

4B04 3E01      DISKRD: MVI     A,DC$SIN        ;SWITCH DD INTO SYSTEM.
4B06 D343              OUT     D$PORT          ;ISSUE DD COMMAND.
4B08 3E01              MVI     A,DC$RDS        ;READ SECTOR COMMAND.
4B0A CD8A4B            CALL    DSK$EX          ;PERFORM OPERATION.
4B0D C24A4B            JNZ     DSK$ER          ;ERROR EXIT.
4B10 2A604C            LHLD    BT$DMA          ;LOAD USER BUF ADDRESS
4B13 EB                XCHG                    ;MOVE HL TO DE.
4B14 018003            LXI     B,DD$BUF        ;LOAD BUFFER OFFSET.
4B17 2A4000            LHLD    D$ADDR          ;LOAD DD WINDOW ADDR.
4B1A 09                DAD     B               ;HL NOW SECTOR BUFFER.
4B1B 018000            LXI     B,SEC$SZ        ;LOAD SECTOR SIZE.
4B1E CD414C            CALL    BLOCK           ;BLOCK MOVE ROUTINE.
4B21 C3444B            JMP     DSK$OK          ;NORMAL RETURN.

               ;*************************************************
```

```
                ; WRITE A DISK SECTOR ROUTINE                          *
                ;*************************************************************

4B24 3E01       DISKWR: MVI     A,DC$SIN        ;SWITCH DD INTO SYSTEM.
4B26 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4B28 018000             LXI     B,SEC$SZ        ;LOAD SECTOR SIZE.
4B2B 2A4000             LHLD    D$ADDR          ;DD SYSTEM ADDRESS.
4B2E 118003             LXI     D,DD$BUF        ;DD BUFFER OFFSET.
4B31 19                 DAD     D               ;HL NOW DD BUF ADDR.
4B32 EB                 XCHG                    ;DE NOW DD BUF ADDR.
4B33 2A604C             LHLD    BT$DMA          ;HL NOW USER BUF ADDR.
4B36 CD414C             CALL    BLOCK           ;BLOCK MOVE ROUTINE.
4B39 3E02               MVI     A,DC$WRS        ;LOAD WRITE SEC CMND.
4B3B CD8A4B             CALL    DSK$EX          ;CALL DISK EXECUTIVE.
4B3E CA444B             JZ      DSK$OK          ;JUMP IF WRITE OK.
4B41 C34A4B             JMP     DSK$ER          ;ERROR EXIT.

                ;*************************************************************
                ; DISK READ/WRITE EXITS                                  *
                ;*************************************************************

4B44 3E00       DSK$OK: MVI     A,DC$SOT        ;SWITCH DD OUT OF SYS.
4B46 D343               OUT     D$PORT          ;ISSSUE HARDWARE CMND.
4B48 AF                 XRA     A               ;ZERO A REGISTER.
4B49 C9                 RET                     ;NORMAL EXIT.

4B4A 3E00       DSK$ER: MVI     A,DC$SOT        ;SWITCH DD OUT OF SYS.
4B4C D343               OUT     D$PORT          ;ISSSUE HARDWARE CMND.
4B4E 3EFF               MVI     A,0FFH          ;LOAD ERROR FLAGS.
4B50 C9                 RET                     ;ERROR EXIT.

                ;*************************************************************
                ; TRANSLATE SECTOR NUMBER                                *
                ;*************************************************************

4B51 7A         SECTRN: MOV     A,D             ;TESTING TBL ADDR.
4B52 B3                 ORA     E               ;ADDR IN REG DE.
4B53 CA5C4B             JZ      NOTRAN          ;IF ZERO, NO TRANS.
4B56 EB                 XCHG                    ;(HL) NOW TRANS TBL.
4B57 09                 DAD     B               ;(HL) NOW TRANS SECTOR.
4B58 6E                 MOV     L,M             ;L IS TRANSLATED SEC.
4B59 2600               MVI     H,0             ;HIGH ORDER BYTE ZERO.
4B5B C9                 RET                     ;RETURN TO CALLER.
4B5C 210100     NOTRAN: LXI     H,1             ;SET HL TO ONE.
4B5F 09                 DAD     B               ;ADD SEC NMBR TO HL.
4B60 C9                 RET                     ;RETURN TO CALLER.

                ;*************************************************************
                ; FORMAT A DISK TRACK ROUTINE                            *
                ;*************************************************************

4B61 3E01       FORMAT: MVI     A,DC$SIN        ;SWITCH DD INTO SYSTEM.
4B63 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4B65 3E03               MVI     A,DC$MB1        ;SELECT DD BANK 1.
4B67 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4B69 010001             LXI     B,FMT$SZ        ;FORMAT PROG SIZE.
4B6C 2A4000             LHLD    D$ADDR          ;DD SYSTEM ADDRESS.
4B6F 110003             LXI     D,DD$FBF        ;DD FORMAT BUF OFFSET.
4B72 19                 DAD     D               ;HL NOW DD FBUF ADDR.
4B73 EB                 XCHG                    ;DE NOW DD FBUF ADDR.
```

```
4B74 2A604C             LHLD    BT$DMA          ;FORMAT PROGRAM ADDR.
4B77 CD414C             CALL    BLOCK           ;BLOCK MOVE ROUTINE.
4B7A 3E01               MVI     A,DC$MBO        ;RESELECT DD BANK 0.
4B7C D343               OUT     D$PORT          ;ISSUE TO DD HARDWARE.
4B7E 3E03               MVI     A,DC$FMT        ;LOAD FORMAT TRK CMND.
4B80 CD8A4B             CALL    DSK$EX          ;CALL DISK EXECUTIVE.
4B83 CD444B             CALL    DSK$OK          ;SWITCH DD MEM OUT.
4B86 3A5F4C             LDA     BT$STS          ;LOAD FORMAT STATUS.
4B89 C9                 RET                     ;FORMAT EXIT.

                ;*********************************************************
                ; DOUBLE D EXECUTION SUBROUTINE                            *
                ;*********************************************************

                ;*******( COMMAND BLOCK TO DOUBLE D AND EXEC )*********

4B8A 32584C     DSK$EX: STA     BT$CMD          ;STORE DCM COMMAND.
4B8D 010700             LXI     B,7             ;NMBR BYTE TO MOVE.
4B90 117003             LXI     D,DD$CBT        ;COMMAND BYTE OFFSET.
4B93 2A4000             LHLD    D$ADDR          ;DD SYS ADDRESS.
4B96 19                 DAD     D               ;HL NOW PTS CMND BLK.
4B97 EB                 XCHG                    ;NOW ADDR IN DE.
4B98 21584C             LXI     H,BT$CMD        ;BIOS CMND BLOCK.
4B9B CD414C             CALL    BLOCK           ;PERFORM BLOCK MOVE.
4B9E 3E02               MVI     A,DC$INT        ;LOAD DD INTERRUPT.
4BA0 D343               OUT     D$PORT          ;ISSUE DD INTERRUPT.

                ;*******( WAIT FOR DOUBLE D HALT )*********************

4BA2 3A4200             LDA     D$HALT          ;LOAD HALT BIT MASK.
4BA5 47                 MOV     B,A             ;MASK IN B REGISTER.
4BA6 DB43       DSK$WT: IN      D$PORT          ;READ DD STATUS.
4BA8 A0                 ANA     B               ;TEST HALT* FLAG.
4BA9 C2A64B             JNZ     DSK$WT          ;TEST UNTIL HALTED.

                ;*******( GET DOUBLE D STATUS )************************

4BAC 3E01               MVI     A,DC$SIN        ;SWITCH DD INTO SYS.
4BAE D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
4BB0 EB                 XCHG                    ;EXCHANGE SRC/DSTN.
4BB1 7E                 MOV     A,M             ;STATUS INTO A REG.
4BB2 12                 STAX    D               ;STORE STATUS BYTE.
4BB3 A7                 ANA     A               ;TEST FOR ERRORS.
4BB4 C9                 RET                     ;RETURN TO CALLER.

                ;*********************************************************
                ; LOG-ON  -   SET DISK PARAMETER BLOCK                     *
                ;*********************************************************

                ;*******( CHECK IF LOG-ON REQUESTED )******************

4BB5 3A644C     LOG$ON: LDA     LOG$RQ          ;CHECK LOG REQUEST.
4BB8 E601               ANI     001H            ;LOG ON BIT TEST.
4BBA C2444B             JNZ     DSK$OK          ;RETURN, NO LOG-ON.

                ;*******( READ IDENTITY SECTOR )**********************

4BBD 22624C             SHLD    DT$PTR          ;STORE DRV TBL PNTR.
4BC0 3E01               MVI     A,DC$SIN        ;SWITCH DD INTO SYS.
4BC2 D343               OUT     D$PORT          ;ISSUE HARDWARE CMND.
```

```
4BC4 3E00           MVI     A,DC$LOG        ;LOAD DCM LOG-ON CMND.
4BC6 CD8A4B         CALL    DSK$EX          ;PERFORM DISK OP.
4BC9 CAD24B         JZ      LOG$CK          ;GO TO LOGON ERROR.
4BCC 210000         LXI     H,0             ;ERROR, BAD LOG ON.
4BCF C34A4B         JMP     DSK$ER          ;BIOS EXIT.
                    ;
                    ;*******( CHECK FOR JADE ID )***************************

4BD2 118003 LOG$CK: LXI     D,DD$BUF        ;DD BUFFER OFFSET.
4BD5 2A4000         LHLD    D$ADDR          ;DD SYS ADDRESS.
4BD8 19             DAD     D               ;HL NOW PNTS BUFFER.
4BD9 11E14C         LXI     D,JADEID        ;DE PNTS BIOS ID.
4BDC 0608           MVI     B,ID$SZE        ;SET LABEL SIZE.
4BDE 1A13   LOG$ID: LDAX    D!      INX D   ;GET LABEL CHARACTER.
4BE0 BE23           CMP     M!      INX H   ;DOES ID SECTOR MATCH.
4BE2 C20F4C         JNZ     LG3740          ;ASSUME DISKETTE 3740.
4BE5 05             DCR     B               ;DECREMENT COUNT.
4BE6 C2DE4B         JNZ     LOG$ID          ;CHECK IF ANOTHER CHR.

                    ;*******( DISKETTE CONTAINS ID )************************

4BE9 CD2E4C         CALL    TRNONE          ;ASSUME DDENS.
4BEC CD364C         CALL    DPB$AD          ;GET DPB ADDR IN DE.
4BEF 01A003         LXI     B,DD$DPB        ;DPB ADDR OFFSET.
4BF2 2A4000         LHLD    D$ADDR          ;DD SYSTEM ADDRESS.
4BF5 09             DAD     B               ;HL NOW AT ID DPB.
4BF6 010F00         LXI     B,DPB$SZ        ;DPB SIZE IN BYTES.
4BF9 CD414C         CALL    BLOCK           ;MOVE INTO DPB.
4BFC 11B103         LXI     D,DD$DDF        ;ID DTA DNS OFFSET.
4BFF 2A4000         LHLD    D$ADDR          ;DD SYSTEM ADDR.
4C02 19             DAD     D               ;HL POINTS FLAGS.
4C03 7E             MOV     A,M             ;LOAD FLAGS.
4C04 E604           ANI     04H             ;TEST DATA DENSITY.
4C06 CC244C         CZ      TR3740          ;IF 0 USE 3740 TRN.
4C09 2A624C         LHLD    DT$PTR          ;RELOAD POINTER.
4C0C C3444B         JMP     DSK$OK          ;EXIT BIOS JUMP.

                    ;*******( ASSUME 3740 DISKETTE )************************

4C0F CD244C LG3740: CALL    TR3740          ;SET SECTOR TRANSLATE.
4C12 CD364C         CALL    DPB$AD          ;SET REGISTER DE.
4C15 010F00         LXI     B,DPB$SZ        ;DPB SIZE IN BYTES.
4C18 217F4C         LXI     H,SD$PBK        ;ADDRESS OF BLK IMAGE.
4C1B CD414C         CALL    BLOCK           ;MOVE INTO DPB.
4C1E 2A624C         LHLD    DT$PTR          ;RELOAD POINTER.
4C21 C3444B         JMP     DSK$OK          ;EXIT BIOS JUMP.

                    ;*******( SET 3740 SECTOR TRANSLATION )*****************

4C24 11654C TR3740: LXI     D,SDTRAN        ;SECTOR TRAN TBL ADDR.
4C27 2A624C         LHLD    DT$PTR          ;ADDR DISK PARA HDER.
4C2A 73             MOV     M,E             ;LOW ORDER ADDR.
4C2B 23             INX     H               ;POINT NEXT BYTE.
4C2C 72             MOV     M,D             ;HIGH ORDER ADDR.
4C2D C9             RET                     ;RETURN TO LOG USER.

                    ;*******( SET NO SECTOR TRANSLATION )*******************

4C2E AF     TRNONE: XRA     A               ;ZERO A REGISTER.
4C2F 2A624C         LHLD    DT$PTR          ;ADDR OF PARA HDER.
```

```
4C32 77                 MOV     M,A             ;ZERO LOW ORDER ADDR.
4C33 23                 INX     H               ;NEXT BYTE.
4C34 77                 MOV     M,A             ;ZERO HIGH BYTE.
4C35 C9                 RET                     ;RETURN TO LOG USER.

                ;*******( GET DRIVE PARA BLK ADDR )*********************

4C36 2A624C     DPB$AD: LHLD    DT$PTR          ;ADDR DISK PARA HDER.
4C39 110A00             LXI     D,10            ;DPB TBL PNTR OFFSET.
4C3C 19                 DAD     D               ;NOW AT DPB PNTR.
4C3D 5E                 MOV     E,M             ;LOW ORDER ADDR.
4C3E 23                 INX     H               ;NEXT BYTE.
4C3F 56                 MOV     D,M             ;HIGH ORDER ADDR.
4C40 C9                 RET                     ;RETURN TO LOG USER.

                ;*************************************************
                ; BLOCK MOVE SUBROUTINE - Z80 LDIR WILL FUNCTION HERE *
                ;*************************************************

4C41 7E23       BLOCK:  MOV     A,M!    INX H   ;GET EACH BYTE.
4C43 1213               STAX    D!      INX D   ;STORE EACH BYTE.
4C45 0B78B1             DCX B!  MOV A,B! ORA C  ;DEC LENGTH (MAX 64K).
4C48 C2414CC9           JNZ     BLOCK!  RET     ;FINISH BLOCK AND RET.

                ;*************************************************
                ; MESSAGE DISPLAY ROUTINE - HL REG POINTS TO STRING   *
                ;*************************************************

4C4C 7E         MSG$OT: MOV     A,M             ;LOAD CHARACTER/BYTE.
4C4D FE24C8             CPI     EOM!    RZ      ;EXIT IF TERMINATOR.
4C50 4FCD0CF0           MOV C,A! CALL   CNS$OT  ;DISPLAY CHARACTER.
4C54 23C34C4C           INX H!  JMP     MSG$OT  ;REPEAT FOR NEXT BYTE.

                ;*************************************************
                ; DOUBLE D - DCM COMMAND BLOCK BUFFER                 *
                ;*************************************************

4C58 00         BT$CMD: DB      0               ;DCM COMMAND.
4C59 00         BT$DRV: DB      0               ;DRIVE NUMBER.
4C5A 00         BT$TRK: DB      0               ;TRACK NUMBER.
4C5B 00         BT$SEC: DB      0               ;SECTOR NUMBER.
4C5C 00         BT$SPO: DB      0               ;SPARE BYTE 0.
4C5D 00         BT$CHR: DB      0               ;LIST CHARACTER.
4C5E 00         BT$MOD: DB      00000000B       ;MODE CONTROLS.
4C5F 00         BT$STS: DB      0               ;COMMAND STATUS.

                ;*************************************************
                ; BIOS VARIABLE STORAGE                               *
                ;*************************************************

4C60 0000       BT$DMA: DW      0               ;SYSTEM TRANSFER ADDR.
4C62 0000       DT$PTR: DW      0               ;DRIVE TABLE POINTER.
4C64 00         LOG$RQ: DB      0               ;LOG ON REQUEST REG.

                ;*************************************************
                ; DOUBLE D - MEMORY ASSIGNMENTS  (40-4F HEX)          *
                ;*************************************************

0040 =          D$ADDR  EQU     0040H   ;DD SYSTEM WINDOW ADDR POINTER.
0042 =          D$HALT  EQU     0042H   ;DD HALT STATUS MASK LOCATION.
```

```
                ;*********************************************************
                ; 3740 FORMAT PARAMETERS ******** CP/M SINGLE DENSITY *
                ;*********************************************************

                ;********( SINGLE DENSITY CP/M SECTORING )***************

4C65 01070D1319SDTRAN:  DB      01H,07H,0DH,13H,19H,05H,0BH,11H,17H,03H
4C6F 090F150208         DB      09H,0FH,15H,02H,08H,0EH,14H,1AH,06H,0CH
4C79 1218040A10         DB      12H,18H,04H,0AH,10H,16H

                ;********( DEFAULT DISK PARAMETER BLOCK )****************

4C7F 1A00       SD$PBK: DW      26          ;SECTORS PER TRACK.
4C81 03                 DB      3           ;BLOCK SHIFT FACTOR.
4C82 07                 DB      7           ;BLOCK MASK.
4C83 00                 DB      0           ;NULL MASK.
4C84 F200               DW      242         ;DISK SIZE - 1.
4C86 3F00               DW      63          ;DIRECTORY MAX.
4C88 C0                 DB      11000000B ;ALLOC 0.
4C89 00                 DB      0           ;ALLOC 1.
4C8A 1000               DW      16          ;CHECK SIZE.
4C8C 0200               DW      2           ;TRACK OFFSET.

                ;*********************************************************
                ; ZERO PAGE IMAGE - BLOCK MOVED TO BASE PAGE             *
                ;*********************************************************

4C8E C3034A     BS$IMG: JMP     BIOS+03H        ;WARM BOOT VECTOR.
4C91 00         IO$IMG: DB      IOBYTE          ;I/O BYTE IMAGE.
4C92 00         DF$IMG: DB      DF$DRV          ;DEFAULT DRIVE IMG.
4C93 C3063C             JMP     BDOS+06H        ;BDOS CALL VECTOR.

                ;*********************************************************
                ; MESSAGES                                               *
                ;*********************************************************

4C96 0D0A0D0A   MSG$SO: DB      CR,LF,CR,LF
4C9A 4A41444520         DB      'JADE COMPUTER PRODUCTS',CR,LF
4CB2 3230               DB      'O' + CPM$NK / 10,'O' + CPM$NK MOD 10
4CB4 4B2043502F         DB      'K CP/M 2.2 DDBIOS2',CR,LF,CR,LF,EOM

4CCB 0D0A434350MSG$LE:  DB      CR,LF,'CCP/BDOS LOAD ERROR',EOM

000A =          LF      EQU     0AH             ;ASCII LINE FEED.
000D =          CR      EQU     0DH             ;CARRAIGE RETURN.
0024 =          EOM     EQU     '$'             ;END OF STRING.
001A =          CNTL$Z  EQU     01AH            ;CONTROL-Z (EOF).

                ;*********************************************************
                ; ID LABEL DEFINITIONS                                   *
                ;*********************************************************

4CE1 4A61646520JADEID:  DB      'JADE DD '      ;ID LABEL.
0008 =          ID$SZE  EQU     $-JADEID        ;LABEL SIZE.

                ;*********************************************************
                ; DRIVE PARAMETER HEADER AREA                            *
                ;*********************************************************
```

```
4CE9 0000       D0$DPH: DW      0               ;SECTOR TRAN TBL.
4CEB 0000               DW      0               ;SCRATCH AREA.
4CED 0000               DW      0               ;SCRATCH AREA.
4CEF 0000               DW      0               ;SCRATCH AREA.
4CF1 004E               DW      DIR$BF          ;DIRECTORY BUFFER.
4CF3 804E               DW      D0$DPB          ;DRIVE PARAM BLK.
4CF5 B04E               DW      D0$CHK          ;DRIVE CHANGE BLK.
4CF7 8F4E               DW      D0$ALL          ;DRIVE ALLOCATION.

                        IF      (1-N$DRVS) SHR 15       ;TEST SIGN BIT.
4CF9 0000000000D1$DPH:  DW      0,0,0,0
4D01 004ED04E           DW      DIR$BF,D1$DPB
4D05 004FDF4E           DW      D1$CHK,D1$ALL
                        ENDIF

                        IF      (2-N$DRVS) SHR 15
                D2$DPH: DW      0,0,0,0
                        DW      DIR$BF,D2$DPB
                        DW      D2$CHK,D2$ALL
                        ENDIF

                        IF      (3-N$DRVS) SHR 15
                D3$DPH: DW      0,0,0,0
                        DW      DIR$BF,D3$DPB
                        DW      D3$CHK,D3$ALL
                        ENDIF

                ;*************************************************
                ; BIOS PROGRAM AREA REMAINING                    *
                ;*************************************************

4E00 =          BIOS$U  EQU     BIOS + K$B      ;BEGIN SCRATCH AREA.
00F7 =          BIOS$F  EQU     BIOS$U - $      ;NUMBER OF BYTES FREE.

                        IF      BIOS$F SHR 15   ;TEST PROG OVERFLOW.
                ERROR   EQU     1/0             ;GENERATE ERROR.
                        ENDIF

4D09            F$AREA: DS      BIOS$F          ;USE UP PROG AREA.

                ;*************************************************
                ; DIRECTORY BUFFER AREA - BEGINNING OF SCRATCH AREA   *
                ;*************************************************

4E00            DIR$BF: DS      SEC$SZ

                ;*************************************************
                ; DRIVE TABLE ENTRY - SIZES                      *
                ;*************************************************

000F =          DPB$SZ  EQU     15              ;PARAMETER BLOCK SIZE.
0021 =          ALL$SZ  EQU     33              ;ALLOCATION INFO SIZE.
0020 =          CHK$SZ  EQU     32              ;CHANGED DISK SCRATCH.

                ;*************************************************
                ; DRIVE TABLES - SCRATCH AREAS                   *
                ;*************************************************

4E80            D0$DPB: DS      DPB$SZ          ;DISK PARAMETER BLOCK.
4E8F            D0$ALL: DS      ALL$SZ          ;DISK ALLOCATION INFO.
```

```
4EB0                D0$CHK: DS      CHK$SZ          ;DISK CHANGED SCRATCH.

                            IF      (1-N$DRVS) SHR 15
4ED0                D1$DPB: DS      DPB$SZ          ;DISK PARAMETER BLOCK.
4EDF                D1$ALL: DS      ALL$SZ          ;DISK ALLOCATION INFO.
4F00                D1$CHK: DS      CHK$SZ          ;DISK CHANGED SCRATCH.
                            ENDIF

                            IF      (2-N$DRVS) SHR 15
                    D2$DPB: DS      DPB$SZ          ;DISK PARAMETER BLOCK.
                    D2$ALL: DS      ALL$SZ          ;DISK ALLOCATION INFO.
                    D2$CHK: DS      CHK$SZ          ;DISK CHANGED SCRATCH.
                            ENDIF

                            IF      (3-N$DRVS) SHR 15
                    D3$DPB: DS      DPB$SZ          ;DISK PARAMETER BLOCK.
                    D3$ALL: DS      ALL$SZ          ;DISK ALLOCATION INFO.
                    D3$CHK: DS      CHK$SZ          ;DISK CHANGED SCRATCH.
                            ENDIF

                    ;*************************************************************
4F20                        END
*
```